

# Network




**TU/e**


Technische Universiteit  
**Eindhoven**  
University of Technology

**Where innovation starts**

# Asynchronous vs. synchronous




Asynchronous Request  
(e.g., asking for a web page)




Socket Connection  
(e.g., chat)

# Asynchronous vs. synchronous



Asynchronous Request  
(e.g., asking for a web page)

Homework




Socket Connection  
(e.g., chat)

# Client/Server

- **Help>Reference: Libraries : Network : Server**
- **Help>Reference: Libraries : Network : Client**

- Before we continue...



- Find your IP
- On windows:
  - WIN+R, cmd, ipconfig
- On Mac:
  - Applications menu>Utilities>Terminal, ifconfig

```
Ethernet adapter Local Area Connection:

Connection-specific DNS Suffix  . : 
Link-local IPv6 Address . . . . . : fe80::c151:61d4:be20:6194%10
IPv4 Address. . . . . : 192.168.105.47
Subnet Mask . . . . . : 255.255.255.0
Default Gateway . . . . . : 192.168.105.1
```

# Client/Server

```
import processing.net.*;
Server myServer;
int val = 0;

void setup() {
    size(200, 200);
    // Starts a myServer on port 5204
    myServer = new Server(this, 5204);
}

void draw() {
    val = (val + 1)%255;
    background(val);
    myServer.write(val);
}
```

```
import processing.net.*;
Client myClient;
int dataIn;

void setup() {
    size(200, 200);
    // Connect to the local machine at port 5204.
    // This example will not run if you haven't
    // previously started a server on this port
    myClient = new Client(this, "127.0.0.1", 5204);
}

void draw() {
    if (myClient.available() > 0) {
        dataIn = myClient.read();
    }
    background(dataIn);
}
```

# Client/Server

- **Now try to change the server code:**
  - **Reacts to mouse clicks**
  - **Position of the mouse changes the background color**
  - **Send the background color to clients**

# Client/Server

```
import processing.net.*;
Server myServer;
int val = 0;

void setup() {
    size(200, 200);
    // Starts a myServer on port 5204
    myServer = new Server(this, 5204);
}

void draw() {
    background(val);
}

void mousePressed(){
    val = mouseY;
    myServer.write(val);
}
```

```
import processing.net.*;
Client myClient;
int dataIn;

void setup() {
    size(200, 200);
    // Connect to the local machine at port 5204.
    // This example will not run if you haven't
    // previously started a server on this port
    myClient = new Client(this, "127.0.0.1", 5204);
}

void draw() {
    if (myClient.available() > 0) {
        dataIn = myClient.read();
    }
    background(dataIn);
}
```

# Client/Server

- **Now try out with your neighbor Ms/r Nice:**
  - Nice runs the server.
  - You replace “127.0.0.1” in your client with the IP address of Ms/r Nice’s computer
  - You run the client.
- Try the opposite.
- Later you can always try the same.

# Client/Server

- **Now let's try the opposite**
- **Now try to change the client code:**
  - **Reacts to mouse clicks**
  - **Position of the mouse changes the background color**
  - **Send the background color to the server**

# Client/Server

```
import processing.net.*;
Server myServer;
Client c;
int val = 0;

void setup() {
    size(200, 200);
    // Starts a myServer on port 5204
    myServer = new Server(this, 5204);
}

void draw() {
    c = myServer.available();
    if(c != null){
        val = c.read();
        background(val);
    }
}
```

```
import processing.net.*;
Client myClient;
int val;

void setup() {
    size(200, 200);
    // Connect to the local machine at port 5204.
    // This example will not run if you haven't
    // previously started a server on this port
    myClient = new Client(this, "127.0.0.1", 5204);
}

void draw() {
    background(val);
}

void mousePressed(){
    val = mouseY;
    myClient.write(val);
}
```

# Client/Server

- **Now let's synchronize the background of all clients and the server.**

# Client/Server

```
import processing.net.*;
Server myServer;
Client c;
int val = 0;

void setup() {
    size(200, 200);
    // Starts a myServer on port 5204
    myServer = new Server(this, 5204);
}

void draw() {
    c = myServer.available();
    if(c != null){
        val = c.read();
        background(val);
        myServer.write(val);
    }
}

void mousePressed(){
    val = mouseY;
    background(val);
    myServer.write(val);
}
```

De

```
import processing.net.*;
Client myClient;
int val;

void setup() {
    size(200, 200);
    // Connect to the local machine at port 5204.
    // This example will not run if you haven't
    // previously started a server on this port
    myClient = new Client(this, "127.0.0.1", 5204);
}

void draw() {
    if(myClient.available()>0){
        val = myClient.read();
        background(val);
    }
}

void mousePressed(){
    val = mouseY;
    myClient.write(val);
}
```